
4.3 Wärmestrahlung 

Abbrand eines Holzbalken-Querschnittes 
 
 
Lernziel 
 
Es wird ein Modell mit Wärmeleitung und Strahlungswärmeaustausch transient berechnet. Zwei 
Varianten mit unterschiedlichen Materialeigenschaften werden verglichen. Im Postprocessing wird mit 
einer Programmierung mit Macros und mit der ANSYS-Parametersprache eine Sonderauswertung 
durchgeführt. 
 
 
4.3.1 Variante 1 
 
4.3.1.1 Aufgabenstellung 
 
Bei der Verwendung von Holzbalken in Bauwerken sind 
besondere Brandvorschriften zu beachten. Festlegungen dazu 
gibt es zum Beispiel in DIN 4102 „Brandverhalten von 
Baustoffen und Bauteilen“ und in ISO 834 „Feuerwider-
standsprüfungen“. Nach einer dieser Vorschriften wird im 
Brandfall eine Erwärmung der Umgebung innerhalb von 10 
Minuten auf T=1073 K (800°C) (linear ansteigend) angenommen. 
Unter dieser Randbedingung ist nachzuweisen, dass für eine 
bestimmte Zeit mindestens noch ein Restquerschnitt des 
Balkens unterhalb einer zulässigen Temperatur von T=873 K 
(600°C) liegt, damit bis zum Räumen des Gebäudes die 
Standsicherheit gewährleistet ist. 
 
In Variante 1 dieses Beispiels ist ein rechteckiger Balken aus 
Holz mit einer Breite von b=150 mm und einer Höhe von h=250 
mm nachzuweisen. Abb. 4.3-1 zeigt den Balken. Die 
Anfangstemperatur beträgt T=273 K (0 °C). 
 
Die Materialwerte sind vorgegeben. Die Wärmeleitfähigkeit 
beträgt λ = 1.70 W / (m K), die Dichte ρ = 900 kg / m3 und die 
spezifische Wärmekapazität cp = 1500 J / (kg K). 
 
Während des Brandes ist ein konvektiver Wärmeübergang mit 
einem Wärmeübergangsbeiwert von α = 250 W / (m2 K) 
zugrunde zu legen. Der Emissionskoeffizient für den Wärmeaus-
tausch über Strahlung ist mit ε=0.4 einzusetzen. 
 
Bei der Ergebnisauswertung soll es möglich sein, zu einem 
bestimmten Zeitpunkt die Fläche aller Elemente aufsummieren zu 
können, deren Temperatur unterhalb einer Grenztemperatur von 
T=873 K (600 °C) liegt. Gesucht ist der Flächenanteil dieser 
Elemente im Verhältnis zur Gesamtquerschnittsfläche. 
 
 
4.3.1.2 Variante 1: Idealisierung 
 
Bei der Idealisierung dieser Aufgabenstellung wird davon 
ausgegangen, dass in Balkenlängsrichtung keine Temperaturunterschiede vorliegen und nur der Bereich 
nachzuweisen ist, der nicht von den Balkenenden beeinflusst ist. Daher wird ein 2-dimensionales Modell 
des Balkenquerschnitts mit dem ANSYS-Element PLANE55 erstellt. Das Element dieses Typs wird in 
der Standardeinstellung verwendet, bei der ein ebenes Problem mit der Ausdehnung von einer 
Längeneinheit (hier: 1 m) in Tiefenrichtung zugrunde liegt. 
 
Es liegen Symmetrien in Bezug auf die 2 Achsen vor, die durch die Querschnittmitte gehen. Daher wird 
nur ein Viertel des Querschnitts modelliert. Abb. 4.3-1 zeigt dieses Viertel, das bei der folgenden 
Dateneingabe zugrunde liegt. 

 

 
Abb. 4.3-1  Aufgabenstellung 

Die für die Konvektion maßgebenden Daten sind die Temperatur der Umgebungsluft und der 
Wärmeübergangskoeffizient. Die Temperatur des umgebenden Mediums beträgt zu Beginn der 
Berechnung T=273 K (0 °C), steigt dann in 10 Minuten linear auf T=1073 K (800 °C) und bleibt 
anschließend auf diesem Wert. 
 
Der Strahlungs-Wärmeaustausch zwischen der Balkenoberfläche und der Umgebung ist in dieser 
Aufgabe einfach zu idealisieren, da der Sicht- bzw. Formfaktor sich für jeden Teil der Balkenoberfläche 
zu 1.0 ergibt. Dieser Wert trifft zu, wenn eine Fläche im Strahlungsaustausch mit dem gesamten 
Halbraum steht, der die Oberfläche überspannt. Im vorliegenden Fall entspricht dies einem Brand, bei 
dem die gesamte Umgebung des Balkens die vorgegebene Umgebungstemperatur angenommen hat 
und nun von jeder Position des umgebenden Raumes Strahlungswärme an den Balken abgegeben wird. 
Für diese Strahlungsbedingung eignet sich bei einer 2-dimensionalen Modellierung besonders der 
ANSYS-Elementtyp SURF151. Dieser Elementtyp stellt ein linienförmiges Oberflächeneffekt-Element 
dar, das am Rand des Modells angeordnet werden kann. Das Element kann den Strahlungsaustausch 
zwischen dem Modellrand und einem Umgebungsknoten abbilden, wobei das Stefan-Boltzmann-Gesetz 
zugrundeliegt. Jedes dieser Elemente ist beschrieben durch 2 Knoten, die an einer Elementkante des 
Modells liegen, und einem weiteren Knoten, der die Umgebung repräsentiert. Dieser Umgebungsknoten 
ist bei der 2-dimensionalen Anwendung zwar an einem Ort in der Ebene angeordnet, die Position ist 
jedoch nicht von Bedeutung. Das Element ist in diesem Fall nichtlinear, da der Strahlungs-Wärme-
austausch nicht von der Differenz der Temperaturen T, sondern von der Differenz der absoluten T4-
Werte abhängt. Ein entsprechender Elementtyp für 3-dimensionale Anwendungen ist der ANSYS-
Elementtyp SURF152. 
 
Die Temperatur des Umgebungsknotens, der den umgebenden Raum repräsentiert, verläuft über die 
Zeit wie bei der Konvektion beschrieben: zu Beginn der Berechnung T=273 K (0 °C), in 10 Minute linear 
auf T=1073 K (800 °C) ansteigend und anschließend auf diesem Wert gleichbleibend. 
 
Es werden SI-Einheiten und für die Temperaturen die Kelvin-Skala verwendet. 
 
 
4.3.1.3 Variante 1: Das Preprocessing 
 
Auswahl des Preprocessor mit 
/PREP7 
 
Der Elementtyp für den Balkenquerschnitt ist hier PLANE55 ohne besondere Optionen (keine 
Rotations- bzw. Axisymmetrie) 
ET,1,PLANE55 
 
Zusätzlich deklarieren wir den Elementtyp, der an der Oberfläche die Strahlungsbedingung zur 
Umgebung abbilden wird, hier SURF151. Für den Einsatz in unserem Beispiel verwenden wir 
besondere Einstellungen dieses Elementtyps, die über die Schalterwerte der "key options" ausgewählt 
werden 
ET,2,SURF151,1,,,1,1 
KEYOPT,2,9,1 
 
Als Zusatzdaten (real constants) verlangt dieser Elementtyp die Vorgabe des Formfaktors sowie der 
Stefan-Boltzmann-Konstante. Wir geben dieser Datengruppe die laufende Nummer 2 und verwenden 
hierfür 
R,2,1.0,5.67e-8 
 
Als Materialverhalten werden die Wärmeleitfähigkeit, die Dichte, die spezifische Wärmekapazität und 
der Emissionsgrad eingegeben 
MP,KXX,1,1.7 
MP,DENS,1,900 
MP,C,1,1500 
MP,EMIS,1,0.40 
 
Die Geometrie des Balkenquerschnitts wird in der x-y-Ebene mit der "top-down"-Methode definiert. Das 
Rechteck, das wir für das abzubildende Viertel festlegen müssen, steht direkt als Standardgeometrie 
(primitive) zur Verfügung. Die Kantenlängen werden als Parameter vereinbart (so dass sie später 

einfacher zu zitieren sind) 
BREITE=150e-3 
HOEHE=250e-3 
und als Abmessungen des Rechtecks eingesetzt 
RECT,0,BREITE/2,0,HOEHE/2 
 
Zur automatischen Vernetzung schreiben wir vor, dass eine 
Elementkantenlänge von 10e-3 m nicht überschritten werden soll 
und dass ausschließlich Rechteckelemente verwendet werden 
sollen 
ESIZE,10e-3 
MSHKEY,1 
MSHAPE,0,2D 
MSHAPE,0,3D 
 
Die Fläche wird vernetzt mit 
AMESH,ALL 
 
Die Elementierung kann grafisch kontrolliert werden mit 
EPLOT 
 
Abb. 4.3-2 zeigt die Anordnung der Elemente. 
 
Die Elemente des Außenrandes, die die Strahlung abbilden, werden als Elemente des Elementtyps 2 
und der Zusatzdaten (real constants) 2 in unserem Beispiel festgelegt. Diese Zuweisungen werden als 
Schalter festgelegt mit 
TYPE,2 
REAL,2 
 
Die Oberflächen, die mit diesen Elementen zu belegen sind, sind durch die Linien 2 und 3 gebildet. 
Vergleichen Sie hierzu die Modellkanten mit 
/PNUM,LINE,1 
LPLOT 
 
Durch die Selektion der genannten Linien und der auf diesen Linien liegenden Punkte mit 
LSEL,S,,,2,3 
NSLL,S,1 
sind die zu verwendenden Oberflächen festgelegt. Zur Vorbereitung der Festlegung der Oberflächen-
elemente wird noch ein Knoten definiert, der im weiteren Verlauf die Umgebungsbedingungen erhält und 
der Ausgangspunkt der Strahlung aus der Umgebung sein wird. Wir 
verwenden für diesen Knoten eine beliebige, zur Zeit freie 
Knotennummer und eine beliebige geometrische Position (denn die 
Strahlungsbedingungen sind durch die Strahlerflächen und den 
Sichtfaktor vollständig festgelegt, der Abstand und die Richtung 
sind nicht mehr ausschlaggebend) 
N,490,BREITE/1.8,HOEHE/1.8 
 
Die entlang dieser Kante zu erzeugenden Elemente werden 
automatisch mit 
ESURF,490 
geschaffen, wobei der Knoten 490 als Umgebungsknoten 
eingetragen wird. 
 
Wir machen die Selektion der Geometriegrößen rückgängig mit 
ALLS 
 
Eine grafische Kontrolle der Elementanordnung erfolgt mit einem 
Plot der Elemente, bei dem zuvor die Nummerierung der 
Elementtyp-Nummern eingeschaltet wird 
/PNUM,TYPE,1 
EPLOT 
 

 
Abb. 4.3-2  Elemente 

 
Abb. 4.3-3  Elemente, 
Typ-Nummern 

Eine noch deutlicher erkennbares Bild wird durch eine verkleinerte 
Darstellung der Elemente erreicht (shrinking) mit 
/SHRINK,.2 
EPLOT 
Utility Menu> Plot Ctrls> Style> Size and Shape 
/SHRINK Shrink entities by 20 percent 
OK 
Utility Menu> Plot> Elements 
 
Bei dieser Darstellung wird jedes Element für sich verkleinert 
dargestellt, wie ein Pullover, der zu heiß gewaschen wurde. 
Dadurch sind die Grenzen der Elemente deutlich erkennbar. 
Besonders die Kombination der ebenen und Linien-Elemente am 
Rand des Modells wird getrennt dargestellt. Stellen Sie danach die 
Verkleinerung wieder zurück mit 
/SHRINK 
 
 
4.3.1.4 Variante 1: Der Lösungsabschnitt 
 
Der Aufruf des Lösungsteils des ANSYS/ED-Programms 
/SOLU 
 
Es wird die transiente Berechnungsmethode ausgewählt 
ANTYPE,TRANS 
 
Als zusätzliche Festlegung, die für alle folgenden Lastfälle gilt, werden die Ergebnisdaten für jeden 
Zeitschritt und in vollem Umfang zur Abspeicherung auf der Ergebnisdatei angefordert mit dem 
Kommando 
OUTRES,ALL,ALL 
 
Hiermit werden insbesondere auch die Resultate von Zwischenschritten (substeps) abgespeichert 
(standardmäßig wird nur der Endzustand eines Lastschrittes und somit nur der letzte Substep 
abgespeichert). 
 
Bei den folgenden Zeitschrittberechnungen soll das ANSYS/ED-Programm innerhalb gewisser, vom 
Anwender vorzugebender Grenzwerte die Schrittweite automatisch auswählen. Diese Zeit-
schrittautomatik ist generell zu empfehlen, da das Programm die erforderlichen und zulässigen 
Schrittweiten aus den aktuellen Ergebnissen und Ergebnisinkrementen gut bestimmen kann und dadurch 
viel Rechenzeit eingespart werden kann. Nur in besonderen Fällen sollte auf die Automatik verzichtet 
werden. Die Zeitschrittautomatik wird aktiviert mit 
AUTOTS,ON 
 
Die Startbedingungen werden nun als Lastschritt 1 eingegeben. Das Ergebnis dieses Lastschrittes ist 
eine Lösung, die die Temperaturverteilung unter den zu Beginn vorliegenden Konditionen darstellt. Im 
vorliegenden Fall ist die Lösung trivial: der gesamte Querschnitt hat T = 273 K. Diese Vorgehensweise, 
hierfür einen gesonderten Lastschritt ablaufen zu lassen, wird hier jedoch dargestellt, da vielfach in 
anderen Anwendungen eine Transiente ausgehend von einer ungleichförmigen Verteilung zu berechnen 
ist. 
 
Bei dieser Vorgehensweise wird mit der folgenden Eingabe die Temperatur angegeben, die zu Beginn 
des ersten Lösungsschrittes zur Bestimmung der nichtlinearen Materialdaten verwendet wird 
TUNIF,273 
Main Menu> Solution> -Loads- Settings> Uniform Temp 
 
Von diesem Wert ausgehend wird in allen Berechnungsschritten jeweils die am Knoten bzw. im Element 
berechnete Temperatur verwendet. Diese uniforme Starttemperatur ist keine Festlegung der 
Temperatur, also nicht zu verwechseln mit einer Randbedingung. Die uniforme Starttemperatur ist nur 
eine Hilfe für den Berechnungsstart. Wenn die Lösung beginnt und bei nichtlinearen, 
temperaturabhängigen Eigenschaften – hier bei uns die Strahlung – eine erste Temperatur gewählt 
werden muss, um die Austauschkoeffizienten mit T4 konkret zu berechnen, dann soll das Programm 
nicht T = 0 einsetzen, sondern T = 273. Das wird die erste Iteration schon „richtiger“ machen, also 
schon zu einem geringeren Residuum (Ungleichgewicht) führen. Dann wird mit weniger Iterationen die 

 
Abb. 4.3-4  Elemente, Typ- 
Nummern, verkleinert (shrink) 

Konvergenz erreicht werden. Diese Starttemperatur ist besonders bei Strahlungsaufgaben sinnvoll, bei 
denen mit absoluten Temperaturen (also Kelvin) gerechnet wird. Ohne 
TUNIF,273 
würde dann in der ersten Iteration mit T=0 gestartet werden, möglicherweise als Lösung eine etwas 
geringere Temperatur als dieser Startwert, also eine negative Temperatur, berechnet werden. Dies wird 
programmintern abgefangen und erzeugt eine Fehlermeldung und einen Abbruch der Lösung, weil 
negative Absoluttemperaturen physikalisch unsinnig sind. Ein solcher Abbruch ist eine lästige 
Behinderung, die durch den Start bei 273 K vermeidbar ist. 
 
Die Konvektion wird an den Knoten der Ränder festgelegt, die geometrisch bei den Koordinaten  
x = breite/2 = 0.150/2 m und y = hoehe/2 = 0.250/2 m liegen. Diese Knoten werden selektiert mit 
NSEL,S,LOC,X,BREITE/2 
NSEL,A,LOC,Y,HOEHE/2 
und (um sie später als Gruppe zitieren und ansprechen zu können) als Komponente unter dem Namen 
“aussen” gespeichert mit 
CM,AUSSEN,NODE 
 
Eine graphische Kontrolle der Knoten, die hiermit erfasst werden, kann erfolgen mit 
/PNUM,NODE,1 
NPLOT 
 
Für den durch diese Knoten dargestellten Rand wird die erzwungene Konvektion eingegeben mit einem 
Wärmeübergangskoeffizienten (HCOEF) von α = 250 W/(m2 K) und einer Umgebungstemperatur 
(TBULK) von TB = 273 K 
ALPHA=250 
SF,ALL,CONV,ALPHA,273 
 
Die Selektion der Randknoten wird aufgehoben mit 
ALLS 
 
Die Temperatur der Strahlungsumgebung wird eingegeben durch 
D,490,TEMP,273 
 
Die Problemzeit für diesen Anfangszustand setzen wir so klein ein, dass noch keine transienten Effekte 
wirksam werden 
TIME,1e-6 
(alternativ kann dies auch über das Ausschalten der transienten Effekte mit TIMINT,OFF erfolgen) und 
lassen die Lösung erstellen mit 
SOLVE 
 
Damit ist die Dateneingabe von Lastschritt 1 abgeschlossen. Die Temperaturverteilung wird berechnet, 
die Eingabe der weiteren Lastschritte kann folgen. 
 
Als Lastschritt 2 wird nun der Zeitabschnitt berechnet, in dem die Temperatur der Umgebung linear 
ansteigt von dem Startwert T=273 K (0 °C) bis zum Endwert von T=1073 K (800 °C). Die folgenden 
Angaben stellen die Werte für den Endpunkt des Zeitabschnittes dar. Für dieses Intervall wird die 
Konvektion entsprechend dem Endwert für die außen liegenden Knoten (diese sind als Komponente 
unter dem Namen “außen” vorher bereits zusammengefasst worden) festgelegt mit 
SF,AUSSEN,CONV,ALPHA,273+800 
 
Die Temperatur der Strahlungsumgebung wird ebenfalls erhöht auf 
D,490,TEMP,273+800 
 
Die Konvektion ist damit für den gesamten Zeitabschnitt definiert durch 

• die Festlegung des vorangegangenen Lastschrittes (der für den aktuellen Lastschritt ja die 
Startbedingungen darstellt), 

• die hier gerade festgelegten Endwerte und 
• einen linearen Verlauf innerhalb des Zeitintervalls, das der aktuelle Lastschritt 2 umfasst. 

Der lineare Verlauf ist standardmäßig vorgesehen. Zur Verdeutlichung wird jedoch das Kommando 
genannt, das diese Einstellung beeinflusst 
KBC,0 
Main Menu> Solution> Load Step Opts> Time/Frequenc> Time and Substeps 

KBC Stepped or ramped b.c.  Ramped 
(diese "Kennziffer für boundary conditions" kann mit 0 auf linearen Verlauf (ramp change) und mit 1 auf 
Verlauf entsprechend einer Sprungfunktion (step change) eingestellt werden). 
 
Der Endpunkt des Zeitabschnittes von 
Lastschritt 2 beträgt 10 Minuten (ein-
gegeben als 10*60 s in der Zeiteinheit 
dieses Beispiels) 
TIME,10*60 
 
Dieser Zeitabschnitt von Lastschritt 2 
(loadstep 2) soll in 60 Zwischenschritten 
(substeps) - also Zwischenschritten von 
jeweils 600/60 s bzw. 10 s - abgearbeitet 
werden. Die Anzahl der Substeps wird als 
Anfangswert von uns vorgegeben. Im 
weiteren Verlauf wird jedoch der Zeit-
schrittautomatik der Spielraum zur 
Anpassung der Substep-Anzahl durch die 
Grenzwerte von Nmax = 60 und Nmin = 20 
gegeben 
NSUBST,60,60,20 
 
Diese Angaben reichen aus, um das zu berechnende transiente Geschehen zu beschreiben. Die Lösung 
erfolgt mit 
SOLVE 
 
Der Konvergenz-Monitor – also die Überwachung des Konvergenzverhaltens bei der Lösung – zeigt am 
Anfang einige Schritte, bei denen sofort Konvergenz erreicht ist und das Ungleichgewicht HEAT kleiner 
ist als das Kriterium HEATCRIT. Bei höheren Temperaturen ist der Einfluss der Strahlung auf die 
Systemmatrix zu erkennen, es treten immer mal Iterationen mit unzulässig hohem Ungleichgewicht auf, 
die eine Korrektur und eine neue Iteration erfordern, bevor Konvergenz erreicht ist und der nächste 
Substep begonnen werden kann. 
 
Damit ist Lastschritt 2 abgeschlossen. 
 
Der nun folgende Lastschritt 3  ist der Zeitabschnitt, in dem die Temperatur der Umgebung den 
Endwert von 1073 K (800 °C) erreicht hat und nicht geändert wird. Die Lastangaben ändern sich also 
nicht mehr, es werden nur einige Optionen des Berechnungsablaufes geändert und das zu berechnende 
Zeitintervall festgelegt. 
 
Der Endpunkt des Zeitabschnittes von Lastschritt 3 beträgt 20 Minuten 
TIME,20*60 
 
Auch dieser Zeitabschnitt von Lastschritt 3 soll in maximal 60 und mindestens 20 Zwischenschritten 
abgearbeitet werden 
NSUBST,60,60,20 
 
Die Lösung erfolgt mit 
SOLVE 
 
Damit ist Lastschritt 3 abgeschlossen. 
 
 
4.3.1.5 Variante 1: 

Das Postprocessing mit dem Zeitverlauf-Postprocessor 
 
Zur Auswertung dieser transienten Berechnung wird der Zeitverlaufs-Postprozessor POST26 mit 
/POST26 
Main Menu> TimeHist Postpro 
aufgerufen. Dieser Postprozessor verarbeitet ausgesuchte Ergebniswerte der Berechnung als 
Zeitverlauf, also als Wertetabelle, bei der jeder Wert dem jeweiligen Zeitpunkt zugeordnet ist. 

 
Abb. 4.3-5  Konvergenz-Monitor 

Standardmäßig wird die Problemzeit als Tabelle 1 eingelesen und bereitgestellt. 
 
Im vorliegenden Beispiel werden die Temperaturen der Knoten in der Mitte der oberen kurzen Kante (an 
der Symmetrielinie), im Zentrum des Querschnitts und in der Mitte der langen rechten Kante (ebenfalls 
an der Symmetrielinie) ausgewertet. Diese Positionen haben die Koordinaten (0,hoehe/2,0), (0,0,0) und 
(breite/2,0,0). Ohne die Nummern der dort angeordneten Knoten zu kennen, können mit den folgenden 
Kommandos die Temperaturen für diese Knoten bezeichnet werden 
NSOL,2,NODE(0,HOEHE/2,0),TEMP,,Oben 
NSOL,3,NODE(0,0,0),TEMP,,Zentrum 
NSOL,4,NODE(BREITE/2,0,0),TEMP,,Seite 
NSOL,5,490,TEMP,,Last 
Main Menu> TimeHist Postpro> Define Variables 
 
Die letzte dieser Eingabezeilen fordert das Einlesen der Temperatur des Umgebungsknotens. 
 
Ehe wir das Diagramm darstellen lassen, 
wird noch vorbereitend festgelegt, dass 
ein Liniengitter unterlegt werden soll 
/GRID,1 
Utility Menu> Plot Ctrls> Style> 
 Graphs> Modify Grid 
und dass die Achsen mit folgenden 
Texten beschriftet werden sollen 
/AXLAB,Y,TEMP [K] 
/AXLAB,X,Zeit [s] 
Utility Menu> Plot Ctrls> Style> 
 Graphs> Modify Curve 
 
Nun lassen wir die berechneten 
Temperaturen an diesen Knoten des 
Modells für alle berechneten Zeitpunkte 
als Diagramm darstellen mit 
PLVAR,2,3,4,5 
Main Menu> TimeHist Postpro> 
 Graph Variables 
 
 
4.3.1.6 Variante 1: 

Das Postprocessing mit dem allgemeinen Postprocessor 
 

Zu dieser Auswertung wird der allgemeine Postprocessor POST1 mit 
/POST1 
aufgerufen. Zunächst kann ein Überblick hergestellt werden, für welche Zeitpunkte Ergebnisdaten 
vorliegen. Dies geschieht mit dem Kommando 
SET,LIST 
 
Die Auswertung von Ergebnissen eines Zeitpunktes - z.B. des letzten berechneten Zeitpunktes - erfolgt 
durch das Einlesen der gewünschten Ergebnisse mit 
SET,LAST 
und die grafische Darstellung der Temperaturverteilung für diesen Zeitpunkt mit 
PLNS,TEMP 
 
Für die Auswertung der Restquerschnittsfläche, die unterhalb der Grenztemperatur liegt, wird hier eine 
Vorgehensweise gewählt, die auf die ANSYS-Parametersprache aufbaut (ANSYS parametric design 
language, APDL). Es wird ein Macro FLAESUM.MAC dargestellt, das die Summation der Flächen 
durchführt. Das Macro kann - wenn es einmal erstellt ist - wie ein eigenes Spezialkommando in der 
interaktiven ANSYS/ED-Programmanwendung benutzt werden. 

 
Abb. 4.3-6  Temperatur-Zeitverlauf 

Das Macro kann im ANSYS/ED-Programm erstellt werden oder in einem beliebigen Editor auf Ihrem 
Rechner geschrieben werden. Bei der Erstellung im ANSYS/ED-Programm müssen die Kommandos 
*CREATE,.. mit dem Namen des Macros (hier gewählt: FLAESUM.MAC) am Anfang des Blockes und 
*END am Ende des Blockes stehen. Bei Verwendung eines Editors entfallen diese beiden Zeilen, die 
geschriebene Datei muss dann unter dem Namen (hier FLAESUM.MAC) abgespeichert werden. 
 
Alle Texte, die auf Ausrufungszeichen (!) folgen, sind Kommentare, die nicht vom ANSYS/ED-
Programm ausgeführt werden. Diese Zeilen können auch weggelassen werden, ohne die Funktion des 
Macros zu beeinträchtigen. 
 
Das Macro ist mit folgenden Zeilen zu erstellen 
*CREATE,FLAESUM,MAC 
/NOPR 
*GET,ZEIT,ACTIVE,,SET,TIME 
FLAGES=0 
EAKT=0 
:A1 
EAKT=ELNEXT(EAKT) 
*IF,EAKT,LE,0,:E1 
N1=NELEM(EAKT,1) 
N2=NELEM(EAKT,2) 
N3=NELEM(EAKT,3) 
N4=NELEM(EAKT,4) 
NANZ=4 
*IF,N4,EQ,0,THEN 
NANZ=3 
*ENDIF 
T1=TEMP(N1) 
T2=TEMP(N2) 
T3=TEMP(N3) 
T4=TEMP(N4) 
TM=(T1+T2+T3+T4)/NANZ 
*IF,TM,LE,TGRENZ,THEN 
*GET,EFLA,ELEM,EAKT,AREA 
FLAGES=FLAGES+EFLA 
*ENDIF 
*GO,:A1 
:E1 
/GOPR 
*MSG,NOTE,FLAGES 
DIE RESTFLAECHE BETRAEGT %G 
*END 
 
Die Funktion des Macro und der einzelnen Kommandos innerhalb des Blockes ist anhand der 
Kommentare sowie mit den Angaben von Band 1 der Reihe „FEM für Praktiker“ zur Erläuterung der 
allgemeinen Kommandos des ANSYS/ED-Programms nachvollziehbar. In kurzen Worten werden im 
Macro folgende Schritte durchgeführt: 

• die Ausgabe im Output-Fenster wird ausgeschaltet (dies spart Zeit) 
• die aktuelle Problemzeit der gerade untersuchten Ergebnisse wird abgefragt, 
• der Wert der Summe der Restflächen flages wird auf Null als Startwert gesetzt, 
• die Nummer des aktuell verarbeiteten Elements wird auf Null als Startwert gesetzt, 
• eine Sprungmarke :a1 wird eingefügt, 
• die auf EAKT als nächsthöhere Elementnummer folgende Elementnummer der im Augenblick 

selektierten Menge der Elemente wird abgefragt, 
• wenn diese Nummer als Null gemeldet wird, wird zu Marke :e1 gesprungen, 
• die Temperaturen der Eckknoten des Elementes EAKT werden abgefragt, 
• der Mittelwert dieser Temperaturen gebildet, 
• dabei wird eine Vorsorge für dreieckige Elemente getroffen. 
• Wenn dieser Mittelwert unterhalb des Grenzwertes tgrenz liegt, wird die Fläche des aktuellen 

Elementes zu der Summe der Restflächen FLAGES hinzuaddiert. 

 
Abb. 4.3-7  Temperaturverteilung, Zeit 1200 

• Anschließend wird vor die Abfrage nach der nächsten Elementnummer (Sprungmarke :a1) 
gesprungen. 

• Am Ende wird die Ausgabe im Output-Fenster wieder eingeschaltet und ein Hinweistext mit 
dem Ergebnis ausgegeben. 

 
Wenn nun das Macro verfügbar ist (also entweder als Ergebnis dieser Eingabe im ANSYS/ED-
Programm oder einer Editor-Eingabe (ohne die Kopfzeile *CREATE,FLAESUM,MAC und die Endzeile 
*END) im aktuellen Verzeichnis abgespeichert ist), dann kann die Anwendung im allgemeinen Post-
processor erfolgen. 
 
Die Auswahl eines Zeitpunktes, an dem die Auswertung erfolgen soll, kann z.B. unter Angabe des 
Lastschrittes (load step) und des Zwischenschrittes (substep) erfolgen mit 
SET,3,8 
 
Hiermit wird Lastschritt 3, Zwischenschritt 8 ausgewählt. Die Problemzeit dieses Schrittes beträgt t = 
800 s, die Temperaturkurve lässt erwarten, dass ein Teil des Balkenquerschnittes unterhalb der 
Grenztemperatur von 273+600 K = 600 °C liegt (vergl. Kurve Zentrum) und ein Außenbereich oberhalb 
dieser Grenztemperatur (vergl. Kurven Oben und Seite). 
 
Die Temperaturverteilung für diesen Zeitpunkt kann 
mit 
PLNS,TEMP 
grafisch dargestellt werden. Zusätzlich können noch 
die Zahlenwerte der Isolinien umgestellt werden auf 
273°, 373°,...,1173° mit 
/CONT,,9,273,100 
PLNS,TEMP 
 
Die Auswertung wird für alle Elemente des 
Balkenquerschnitts durchgeführt, die zum Element-
typ 1 gehören 
ESEL,S,TYPE,,1 
 
Der Temperaturgrenzwert wird vorgegeben mit 
TGRENZ=600+273 
und das Macro aufgerufen über 
FLAESUM 
 
Nach dem Ablauf des Macros kann die Fläche 
abgelesen werden. Ebenso können die aktuellen 
Werte der verwendeten Parameter abgefragt 
werden, z.B. der Wert des Parameters FLAGES, 
der die Summe der Flächen aller Elemente enthält, 
die eine mittlere Temperatur kleiner oder gleich der Grenztemperatur haben 
*STATUS,FLAGES 
 
Im vorliegenden Beispiel beträgt diese Restfläche 7.48197E-03 = 0.00748 m2. Die Ausgangsfläche des 
Querschnitts betrug 0.009375 m2. 
 
Die Auswertung des Zeitpunktes t = 1200 s ergibt mit 
SET,3,22 
ESEL,S,TYPE,,1 
TGRENZ=600+273 
FLAESUM 
eine Restfläche von 6.67067E-03 = 0.00667 m2. 
 
An diesem Beispiel soll zusätzlich gezeigt werden, wie ein Macro (hier die Restflächenberechnung mit 
der Eingabe FLAESUM.MAC) innerhalb eines weiteren Macros verwendet werden kann und so die 
eigenen Prozeduren in mehreren Ebenen geschachtelt werden können. 
 
Wenn das Macro FLAESUM.MAC verfügbar ist, kann mit der folgenden Kommandofolge automatisch 
für jeden Zeitpunkt des Ablaufes der transienten Berechnung die Restfläche berechnet und grafisch 

 
Abb. 4.3-8  Temperaturverteilung, Zeit 800 

aufgetragen werden. Dies kann durch eine Schleife programmiert werden, so dass automatisch die 
Folge der Lastschritte ausgewertet und das Macro FLAESUM angewendet wird. Dieser Ablauf sollte 
dadurch nachvollzogen werden, dass die folgenden Zeilen im Editor eingegeben werden und unter dem 
Namen ALLE.MAC als Datei abgespeichert werden 
*CREATE,ALLE,MAC 
/NOPR 
TGRENZ=600+273 
! DER LETZTE ZEITPUNKT 
SET,LAST 
*GET,LZZT,ACTIVE,,SET,TIME 
*DIM,EAR,TABLE,200 
! DER ERSTE ZEITPUNKT 
ESEL,S,TYPE,,1 
N=1 
SET,FIRST 
FLAESUM 
EAR(N)=FLAGES 
! DIE SCHLEIFE 
:A2 
SET,NEXT 
*GET,ZEIT,ACTIVE,,SET,TIME 
*IF,ZEIT,GE,LZZT,:E2 
N=N+1 
ESEL,S,TYPE,,1 
FLAESUM 
EAR(N)=FLAGES 
*GO,:A2 
:E2 
*VLEN,N 
*VPLO,,EAR(1) 
*END 
 
Der erste Abschnitt dieser Eingabe legt einige Ausgangswerte fest und wertet den ersten nach der 
Berechnung abgespeicherten Lastzustand aus. Der zweite Abschnitt enthält eine Rücksprunglogik, mit 
der ein Lastzustand nach dem anderen ausgewertet wird, jeweils das Macro zur Berechnung der 
Restfläche FLAESUM verwendet wird und der Ergebniswert FLAGES in ein Parameter-Array EAR 
eingetragen wird. Am Ende der Eingabe wird das Parameter-Array EAR grafisch dargestellt. Diese 
Eingabefolge wird abgearbeitet mit der Eingabe 
alle 
 
Die Ergebnisdarstellung ist in Abb. 4.3-9 gezeigt. Die Abb. 4.3-9 zeigt einen stufigen Verlauf, da in der 
Auswertung der Restfläche jeweils die mittlere Temperatur jedes Elementes abgefragt wird. Sobald 
dieser Wert die Grenztemperatur überschreitet, wird die gesamte Fläche dieses Elementes aus der 
Restfläche entfernt. Um den Verlauf kontinuierlicher zu berechnen, ist bei dieser Auswertungslogik eine 
Netzverfeinerung erforderlich. 
 
 

 
Abb. 4.3-9  Zeitverlauf Restfläche 

4.3.2 Variante 2 
 
4.3.2.1 Variante 2: Aufgabenstellung 
 
In Variante 2 soll unter gleichen Randbedingungen ein Balken aus anderem Material untersucht werden. 
Mit gleichen Außenabmessungen wird ein Doppel-T-Profil aus Stahl zugrunde gelegt, bei dem die Kehle 
mit Beton gefüllt ist. Die Abmessungen sind wie vorher 150 mm und 250 mm. 
 
Die Materialwerte sind 

• für den Stahl eine Wärmeleitfähigkeit von λ = 40 W / (m K), eine Dichte von ρ = 7850 kg / m3 
und eine spezifische Wärmekapazität von cp = 500 J / (kg K), 

• für den Beton eine Wärmeleitfähigkeit von λ = 10 W / (m K), eine Dichte von ρ = 2000 kg / m3 
und eine spezifische Wärmekapazität von cp = 1000 J / (kg K). 

 
Der Wärmeübergangsbeiwert zur Umgebung bleibt α = 
250 W / (m2 K), der Emissionsgrad für den Wärmeaustausch über 
Strahlung 0.4. 
 
 
4.3.2.2 Variante 1: Die Idealisierung 
 
Die Idealisierung folgt den gleichen Grundsätzen wie Variante 1. 
Auch hier wird nur ein Viertel des Querschnitts modelliert. Abb. 4.3-
10 zeigt dieses Viertel, das bei der folgenden Dateneingabe 
zugrunde liegt. 
 
Die Elementierung wird wie bei Variante 1 durchgeführt. Nach dem 
automatischen Erzeugen der Elemente werden jedoch die 
Materialkennungen für die Elemente des Betonquerschnitts 
umgestellt. 
 
Auch die Lasten bleiben ungeändert: die Temperatur des um-
gebenden Mediums und die Strahlungstemperatur der Umgebung 
beträgt zu Beginn der Berechnung 273 K (0 °C), steigt dann in 10 
Minuten linear auf 1073 K (800 °C) und bleibt anschließend auf 
diesem Wert. 
 
Es werden SI-Einheiten und für die Temperaturen die Kelvin-Skala 
verwendet. 
 
 
4.3.2.3 Variante 2: Das Preprocessing 
 
Die Eingabedaten entsprechen weitgehend denjenigen der Variante 1. Daher werden hier nur die 
Eingabe-Kommandozeilen aufgelistet, die zur Berechnung verwendet wurden. 
/PREP7 
! ELEMENTTYPEN 
ET,1,PLANE55 
ET,2,SURF151,1,,,1,1 
KEYOPT,2,9,1 
! QUERSCHNITTSWERTE 
R,2,1.0,5.67E-8 
! MATERIALDATEN STAHL 
MP,KXX,1,40. 
MP,DENS,1,7850. 
MP,C,1,500. 
MP,EMIS,1,.40 
! MATERIALDATEN BETON 
MP,KXX,2,10. 
MP,DENS,2,2000. 
MP,C,2,1000. 
MP,EMIS,2,.40 

 
 
Abb. 4.3-10  Aufgabenstellung 

! GEOMETRIE 
! BALKENQUERSCHNITT 
BREITE=150E-3 
HOEHE=250E-3 
RECT,0,BREITE/2,0,HOEHE/2 
ESIZE,10E-3 
ESHAPE,2 
TYPE,1 
MAT,1 
REAL,1 
AMESH,ALL 
 
! AUSSENRAND 
TYPE,2 
REAL,2 
LSEL,S,,,2,3 
NSLL,S,1 
N,490,BREITE/1.8,HOEHE/1.8 
ESURF,490 
ALLS 
! AENDERN DER KENNUNG FUER 
! BETONELEMENTE 
NSEL,S,LOC,X,20E-3,999 
NSEL,R,LOC,Y,-1,95E-3 
ESLN,S,1 
ESEL,R,TYPE,,1 
EMOD,ALL,TYPE,1 
EMOD,ALL,MAT,2 
ALLS 

 
Eine grafische Kontrolle der Elementyp-Zuweisung kann ausgeführt werden mit 
/PNUM,TYPE,1 
/SHRINK,.2 
EPLOT 

und eine Kontrolle der Material-Zuweisung (Abb. 4.3-11) mit 
/PNUM,MAT,1 
EPLOT 
 
4.3.2.4 Variante 2: Der Lösungsabschnitt 
 
In diesem Abschnitt der Dateneingabe ergibt sich keine Änderung. Die verwendeten Kommandos 
hierfür lauteten 
/SOLU 
ANTYP,TRANSIENT 
OUTRES,ALL,ALL 
AUTOTS,ON 
 
! LASTFALL 1 
 
! STATIONAERER ANFANGSZUSTAND 
TUNIF,273 
NSEL,S,LOC,X,BREITE/2 
NSEL,A,LOC,Y,HOEHE/2 
CM,AUSSEN,NODE 
! UEBERGANGSKOEFFIZIENT 
ALPHA=250. 
SF,ALL,CONV,ALPHA,273 
ALLS 
! TEMPERATUR DER 
! STRAHLUNGSUMGEBUNG 
D,490,TEMP,273 

 
Abb. 4.3-11  Elemente, Typ- 
Nummern, verkleinert (shrink) 

TIME,1E-6 
SOLVE 
 
! LASTFALL 2 
! TRANSIENTE 
SF,AUSSEN,CONV,ALPHA,273+800 
D,490,TEMP,273+800 
KBC,0 
NSUBST,60,60,20 
TIME,10*60 
SOLVE 
 
! LASTFALL 3 
! TRANSIENTE 
TIME,20*60 
SOLVE 
 
 
4.3.2.5 Variante 2: 
  Das Postprocessing 
 
Im allgemeinen Postprocessor wird die Verteilung 
der Temperaturen am Ende des Berechnungs-
zeitraumes dargestellt mit 
/POST1 
SET,LAST 
/CONT, 
ESEL,S,TYPE,,1 
NSEL,U,,,490 
/UDOC,1,CNTR,LEFT 
PLNS,TEMP 
 
Es ist zu erkennen, dass durch die hohe Leitfähigkeit des Stahles die Mitte des Profils relativ schnell 
warm wird. Dadurch wird die Tragfähigkeit des Stahlquerschnittes verringert. 
 
Im Postprocessing wird der Zeitverlauf der Balkeneckpunkte grafisch aufgezeichnet. Die Punkte, die 
dafür abgefragt werden, sind wie bei Variante 1 das Zentrum, die Außenseiten seitlich und oben und 
zusätzlich die Lastfunktion, also die außen anliegende Umgebungstemperatur. Das Diagramm wird 
dargestellt mit 
/POST26 
NSOL,2,NODE(0,HOEHE/2,0),TEMP,,OBEN 
NSOL,3,NODE(0,0,0),TEMP,,ZENTRUM 
NSOL,4,NODE(BREITE/2,0,0),TEMP,,SEITE 
NSOL,5,490,TEMP,,LAST 
/GRID,1 
/AXLAB,Y,TEMP [K] 
/AXLAB,X,ZEIT [S] 
PLVAR,2,3,4,5 
 
 
 
 

 
Abb. 4.3-12  Temperaturverteilung, Zeit 1200 

 
 
Abb. 4.3-13  Temperatur-Zeitverlauf 

